Parallelportansteuerung unter VB II

Nutzen

Mit dem Parallelport des PCs ist es selbstverständlich nicht nur möglich, externe Geräte anzusteuern, sondern auch, Daten von diesen zu lesen.

Wie, das erfahrt ihr in diesem Tutorial!

Viel Spass

Philipp „Phip“ Burch

Grundlagen

Zuerst noch einige allgemeine Informationen zum LPT, die Sie vielleicht allerdings schon aus dem ersten Tutorial von Julian Habigt kennen:

Die Pinbelegung

Gleich vorneweg: Diese Datei stammt von Stefan Maag, nicht von mir! Ich finde, der Text vermittelt ein sehr gutes Hintergrundwissen, wenn man mit dem LPT arbeiten will. Die komplette Datei finden Sie am Ende des Tutorials.

 [code]

===========================================================================

   Zuordnung LPT-Port-Pins zu den Register-Bits

 ===========================================================================

 Register 0 = DatenRegister   : Adr 378

 Register 1 = StatusRegister  : Adr 379

 Register 2 = ControlRegister : Adr 37A

 BASE ADDRESS  (PDR) Data register (Read or Write)

  Bit 0 - Pin 2 - Data Output 1

  Bit 1 - Pin 3 - Data Output 2

  Bit 2 - Pin 4 - Data Output 3

  Bit 3 - Pin 5 - Data Output 4

  Bit 4 - Pin 6 - Data Output 5

  Bit 5 - Pin 7 - Data Output 6

  Bit 6 - Pin 8 - Data Output 7

  Bit 7 - Pin 9 - Data Output 8

 BASE ADDRESS + 1     (PSR) Status register (Read Only)

  Bit 0 -        - Not used

  Bit 1 -        - Not used

  Bit 2 -        - Not used

  Bit 3 - Pin 15 - ERROR Input

  Bit 4 - Pin 13 - SELECT Input

  Bit 5 - Pin 12 - PAPER END Input

  Bit 6 - Pin 10 - ACKNOWLEDGE Input

  Bit 7 - Pin 11 - BUSY Input

 BASE ADDRESS + 2     (PCR) Control register (Read or Write)

  Bit 0 - Pin  1 - DATA STROBE Output

  Bit 1 - Pin 14 - AUTO LINE FEED Output

  Bit 2 - Pin 16 - INITIALIZE PRINTER Output

  Bit 3 - Pin 17 - SELECT INPUT Output

  Bit 4 -        - 0 = disable interrupt, 1 = enable interrupt

  Bit 5 -

  Bit 6 -

  Bit 7 –

[/code]

Jetzt sehen wir sofort, dass sich unsere gesuchten Anschlüsse an den Pins 10 – 13 und 15 bzw. in den Bits 3 – 7.

Wenn man die Bezeichnungen der Anschlüsse genauer betrachtet, dann sieht man sofort, dass der LPT sehr eng mit dem Drucker verbunden ist. Vor allem PAPER END deutet natürlich sehr auf einen Drucker hin, wodurch der LPT wahrscheinlich auch umgangssprachlich als „Druckerport“ bezeichnet wird.

Zusammenfassung:

Wir wissen, dass die für uns interessanten Anschlüsse die Pins 10, 11, 12, 13 und 15 sind und diese wiederum mit den Bits 3 – 7 angesprochen werden. Ebenso ist bekannt, dass sich der Port an der Adresse Base + 1, also im Normalfall &H379 befindet.

Etwas anderes noch, was schon häufig gefragt wurde:

Kann ich die Anschlüsse des DATA-Registers auch als Eingänge nutzen?
Antwort: Ja, kann man. Allerdings nicht an Address + 0 sondern an Address + 4, also normalerweise &H37C.

Hardware

Um jetzt überhaupt etwas zu machen, ist es natürlich erforderlich, dass wir auch eine Möglichkeit für Eingaben haben. Am geeignetesten dazu sind ganz einfache Taster, die man wie folgt anschliessen sollte:

[image: image1.png]Port-Diagram =[01x]
Eingings
PortfE7s | Itevalk 50| Stat | [CEEE] I X012

Zeit 048.3205 Reset | Mauszeit 0530205
FPS: 020 SPF. 0083

Ausginge

Fort #H[378

Mz e s e M3 C2 1 RO

I Binéizshler [~ Lauficht [~ Birken [ Zufdlia




Pin 15 (ERROR)


Pins 18 – 25 (Masse)

[/box]

Falls sich nichts anderes finden liesse, könnte man es zu Testzwecken natürlich auch einfach mit Drahtbrücken machen.

Wenn man das für jeden der Eingänge einmal aufgebaut hat, dann kann’s ja losgehen (.

Einlesen einzelner Bits

Wer hätte schon gerne mal mehr Buttons am PC gehabt? Zum Beispiel einer, der den Monitor automatisch in den Standby-Modus versetzt, um Strom zu sparen. Warum also nicht? Was brauchen wir dazu?

1. Einen Taster

2. Den LPT

3. Etwas Kabel

4. Ein entsprechendes Programm

5. ... Das wär’s dann eigentlich ja schon.

Die Funktion „Inp“

In der InpOut32.dll hat es zwei Prozeduren:

Die Sub „Out32“ zum Ausgeben von Signalen und

die Funktion „Inp32“ zum Einlesen von Signalen.

Hier interessiert uns natürlich die Funktion „Inp32“. Die Deklaration sieht meistens folgendermassen aus:

[code]

Private Declare Function Inp Lib "inpout32.dll" Alias "Inp32" ( _

  ByVal PortAddress As Integer) As Integer

[/code]

Der Parameter „PortAddress“ sollte eigentlich selbsterklärend sein; er beinhaltet die Adresse des zu lesenden Ports. Diese ist beim LPT wie schon gesagt meistens &H379 für die Eingänge.

Unser erstes Programm mit „Inp“

So, jetzt geht’s endlich an die Programmierung ;). Fangen wir einmal ganz einfach an:

Wir öffnen VB und laden ein neues Projekt. Auf die Form setzen wir erst mal einen Timer („Timer1“) und ein Label („Label1“). In die Form_Load-Prozedur schreiben wir folgendes:

[code]

Private Sub Form_Load()

    Timer1.Interval = 50

    Timer1.Enabled = True

End Sub

[/code]

Das sollte eigentlich einleuchten. Zuerst machen wir einen Intervall von 50ms (Das schnellste, was der VB-Timer noch zustande bringt...) und aktivieren den Timer.

Dann wär’ jetzt also noch das eigentlich Einlesen dran:

[code]

Private Sub Timer1_Timer()

    Label1.Caption = Inp(&H379)

End Sub

[/code]

Auch das ist meiner Meinung nach recht verständlich. Bei jedem Timer-Aufruf wird der aktuelle Wert vom LPT gelesen und im Label ausgegeben. Falls Sie den Port an einer anderen Adresse haben, müssen Sie „&H379“ natürlich anpassen. Die richtige Adresse finden Sie in der Systemsteuerung, im Geräte-Manager.

Jetzt starten wir das Programm und siehe da: Das erste Lebenszeichen des LPTs (.

Damit ist jetzt also der Grundstein für die weitere Programmierung gelegt. Der angezeigte Wert sollte sich eigentlich ändern, wenn Sie den eingebauten Taster drücken...

Nur... Was soll jetzt diese Zahl? Wir haben doch erfahren, dass es bei den digitalen Eingängen nur High und Low, also 1 und 0 gibt...

Wir erinnern uns: Wie wird noch mal etwas auf den LPT ausgegeben? Mittels einer Zahl zwischen 0 und 255, nicht? Aber bei den Ausgängen gibt’s doch auch nur 1 und 0...

Richtig! Es ist eine Zahl, die den Status der acht Bits des Registers anzeigt.

Leider sagt uns diese Zahl jetzt nicht so viel... Wir möchten lieber alle Leitungen einzeln sehen.

Daher müssen wir den Wert ins Binärsystem konvertieren.

Dezimal ( Binär

Vielleicht wissen Sie bereits, wie man dezimale Werte ganz einfach ins Binärsystem konvertieren kann. Dann können Sie diesen Abschnitt selbstverständlich überspringen!

Zuerst einmal betrachten wir den Aufbau einer Binärzahl: (Ich gehe jetzt natürlich immer von den üblichen acht Bits aus!)

Sie besteht aus 8 Zahlen, genauer aus 8 Einsen und Nullen. Diese Zahlen können wir jetzt eigentlich einfach anstatt als 1 und 0 als 9 und 0 ansehen, dann wird es gleich viel einfacher zum Verstehen.

Also:

00101101 ( 00909909

Jetzt können wir mal damit rechnen, z.B. einfach mal Plus 1.

In der schriftlichen Form:

  00909909

+ 00000001

  00909910
Denn Neun plus Eins gibt ja bekanntlich Zehn. Allerdings haben wir die Einsen in Neun umgewandelt, deshalb ist das Ergebnis

00909990
Das wär’s jetzt eigentlich schon. Wir sehen also, dass es beim Binärsystem im Grunde genommen nur die Zahlen 0 und 9 gibt, wobei 0 + 1 natürlich dann auch 0 + 9 wäre, was ja korrekt wieder 9 ergibt. Es ist also klar, dass im Binärsystem einfach mit der Basis 2 gearbeitet wird, denn 0 und 1 sind ja zwei Möglichkeiten.

Das war jetzt einfach Mal eine kleine Einführung in die binäre Zählweise.

Zur Umrechnung benötigen wir jetzt allerdings die sog. logischen Operatoren. Dazu zählen

And (Und)

Or (Oder)

XOr (exklusives Oder) und

Not (nicht)

Mit diesen Operatoren werden nun einzelne Bits miteinander verknüpft.

Bei And:

    00101101

And 01010101

    00000101
Hier werden also die einzelnen Bits per Und verknüpft. Das heisst, dass es bei zwei Nullen eine Null ergibt, bei einer Eins und einer Null eine Null und bei zwei Einsen eine Eins.

Das kann man nun in einer sog. Wahrheitstabelle darstellen:

And

A  B  Q

0  0  0

1  0  0

0  1  0

1  1  1

Für die anderen Operatoren hier auch noch die Wahrheitstabellen:

Or

A  B  Q

0  0  0

1  0  1

0  1  1

1  1  1

XOr

A  B  Q

0  0  0

1  0  1

0  1  1

1  1  0

Not

A  B  Q

0  -  1

1  -  0

Da wir wissen, dass das Binärsystem die Basis Zwei hat, können wir jetzt auf die einzelnen Bits zugreifen. Denn die Zahl 2x liefert immer das X-te Bit einer Zahl, wobei das Bit ganz rechts, das niederwertigste Bit, die Nummer X = 0 hat.

Wenn wir jetzt also die Zahl 00010000 haben möchten, dann schreiben wir einfach 24, da das vierte Bit von rechts eine Eins ist (Ganz rechts ist das nullte Bit!).

Überlegen wir also einmal...

Wir wissen, wie wir ein einzelnes Bit setzen können und wir kennen die logischen Grundoperatoren.

Also können wir den Wert des dritten Bits folgendermassen errechnen:

If 15 And 2 ^ 3 > 0 Then

    Bit = 1

Else

    Bit = 0

End If
Die Zahl 15 hat den binären Wert 00001111, also

    00001111

And 00001000

    00001000
Somit ist das Ergebnis grösser als Null, also hat das Bit eine Eins.

Um jetzt alle acht Bits einer Dezimalzahl herauszubekommen können wir folgenden VB-Code verwenden:

Private Sub ShowBits(Number As Integer)

    Dim i As Long

    For i = 0 To 7      'Die acht Bits

        If (Number And 2 ^ i) > 0 Then

            lblBit(i).Caption = "1"

        Else

            lblBit(i).Caption = "0"

        End If

    Next i

End Sub

Dazu benötigen wir allerdings jetzt kein Label1 mehr, aber dafür acht Labels „lblBit“ mit den Indizes 0 – 7.

Wenn wir das nun in unseren bisherigen Code einbauen, dann sieht er so aus:

[code]

Option Explicit

Private Declare Function Inp Lib "inpout32.dll" Alias "Inp32" ( _

  ByVal PortAddress As Integer) As Integer

Private Sub Form_Load()

    Timer1.Interval = 50

    Timer1.Enabled = True

End Sub

Private Sub Timer1_Timer()

    Call ShowBits(Inp(&H379))

End Sub

Private Sub ShowBits(Number As Integer)

    Dim i As Long

    For i = 0 To 7      'Die acht Bits

        If (Number And 2 ^ i) > 0 Then

            lblBit(i).Caption = "1"

        Else

            lblBit(i).Caption = "0"

        End If

    Next i

End Sub
[/code]

Damit hätten wir jetzt den Wert aller acht Bits vom Eingang. Es sollte allerdings noch erwähnt werden, dass uns die Bits 0 – 2 nicht zu interessieren brauchen, da diese unbenutzt sind.

Da stimmt doch was nicht!

Lassen wir das Programm mal laufen. Ok, die ersten drei Bits interessieren uns nicht, da sie nicht benutzt sind. Aber was ist mit dem Rest? Warum hat das letzte Bit eine Null, obwohl der Anschluss keine Verbindung zur Masse (= Low-Zustand) hat? Sobald man den Schalter allerdings schliesst, wird eine Eins angezeigt... Hmm...

Woran könnte das liegen?

Die Antwort: Es liegt daran, dass das siebte Bit invertiert, also umgekehrt ist. Das können wir jetzt allerdings natürlich nicht brauchen, wir wollen doch den „echten“ Wert der Anschlüsse.

Deshalb müssen wir den Rückgabewert von Inp erst mal noch etwas verändern. Wie gesagt, wir wollen das letzte Bit umkehren. Der Wert davon beträgt 27, also 128. Wir wollen allerdings nur dieses eine Bit umkehren, daher fällt der Not-Operator sowieso weg. Geht es mit And? Nein! Damit wären dann alle anderen Bits Null, da 128 binär 10000000 beträgt. Or ist es auch nicht, da dann das siebte Bit in jedem Fall eine Eins hat. Daher bleibt also nur noch eine Möglichkeit:

XOr

Damit klappts! Wegen dem Or bleiben alle anderen Bits unverändert, nur das siebte wird umgekehrt, da zwei Einsen eine Null ergeben und unterschiedliche Werte eine Eins.

Somit brauchen wir also nur die eine Zeile mit dem Aufruf von ShowBits zu verändern:

[code]

Call ShowBits(Inp(&H379) Xor 128)   'Den Wert des siebten Bits umkehren!

[/code]

Das „ButtonUp“-Ereignis

Von VB sind wir uns ja gewöhnt, dass es für alles ein Ereignis gibt. Für Mausklicks, Tastenanschläge, Grössenveränderungen, ... Sowas gibt’s für den LPT natürlich (noch) nicht. Deshalb schreiben wir das nun selbst. Wir brauchen also eine Variable, bzw. ein Array, in dem die aktuellen Tastenstaten gespeichert sind. Wir nennen ihn mal „ButtonState(4)“ und deklarieren ihn als Boolean.  Um jetzt ein Ereignis auszulösen, wenn der Taster losgelassen wird, müssen wir bei jedem Timer-Aufruf testen, ob das jeweilige Bit eine Eins hat und in ButtonState(i) eine Null steht. Wir wollen das Ereignis ja nicht jedes Mal haben, wenn der Timer gefeuert wird und der Taster losgelassen ist.

Deshalb schreiben wir die Timer1_Timer-Prozedur folgendermassen um:

[code]

Private Sub Timer1_Timer()

    Dim i As Long

    Call ShowBits(Inp(&H379) Xor 128)   'Den Wert des siebten Bits umkehren!

    'Gegebenenfalls ein ButtonUp-Ereignis auslösen

    For i = 0 To 4

        'Testen, ob das jeweilige Bit den Wert Eins hat (= Taste nicht gedrückt)

        If (Inp(&H379) And 2 ^ i) > 0 Then

            'Ist der Button auch wirklich LOSGELASSEN worden?

            If ButtonState(i) Then

                'Ja

                Call ButtonUp(i)

            End If

            'ButtonState() aktualisieren

            ButtonState(i) = False

        Else

            ButtonState(i) = True

        End If

    Next i

End Sub
[/code]

Natürlich brauchen wir jetzt auch noch eine ButtonUp-Prozedur:

[code]

'Zum Testen geben wir hier jetzt einfach mal einen Piep aus

Private Sub ButtonUp(Index As Long)

    Call Beep

End Sub

[/code]

Und siehe da! Es funktioniert! Wenn wir einen Taster drücken und wieder loslassen, dann piept’s (Sofern die Lautsprecher eingeschaltet sind und der PC einen Beep ausgeben kann natürlich!).

Der Standby-Modus

Um den Monitor in den Standby-Modus zu versetzen hab’ ich den Code aus !tip0048 verwendet.

Erstmal brauchen wir also ein Modul. Projekt ( Modul hinzufügen ( Modul

Ins Modul setzen wir dann folgenden Code:

[code]

Option Explicit

Private Declare Function SendMessage Lib "user32" Alias _

        "SendMessageA" (ByVal hwnd As Long, ByVal wMsg As _

        Long, ByVal wParam As Long, lParam As Any) As Long

'Update am 6. April 2003: der Monitor sollte sich nun auch

'unter Windows 2k/XP abschalten

Private Const WM_SYSCOMMAND = &H112&

Private Const SC_MONITORPOWER = &HF170&

Dim CurrentStandby As Boolean   'True, wenn sich der Moni gerade im

                                'Standby-Modus befindet

Public Sub SwitchMonitor()

    If CurrentStandby Then

        SendMessage Form1.hwnd, WM_SYSCOMMAND, SC_MONITORPOWER, ByVal -1&

        CurrentStandby = False

    Else

        SendMessage Form1.hwnd, WM_SYSCOMMAND, SC_MONITORPOWER, ByVal 2&

        CurrentStandby = True

    End If

End Sub
[/code]

Und jetzt noch im ButtonUp-Ereignis die SwitchMonitor-Methode aufrufen und das Prog ist fertig!

Ein Digital-Oszilloskop

Ja, das wär’ doch jetzt noch ein praktischer Abschluss!

Gleich vorneweg: Es ist gar nicht schwer! Ohne die Probleme mit dem Timer hätte ich das Programm nach ein paar Stunden schon fertig gehabt. Also erschrecken Sie nicht vor dem Titel!

Zuerst mal ein Screenshot des fertigen Programms:




























Wie Sie sehen, wird hier gerade ein Binärsignal protokolliert, welches ebenfalls vom Programm kommt, also nicht vom LPT. Aber es wird trotzdem der Umweg über die InpOut32.dll gemacht: Das Signal wird mit Out() ausgegeben und mit Inp() wieder eingelesen. Selbstverständlich könnte man damit aber auch protokollieren, was so auf der Schnittstelle passiert, wenn man etwas ausdruckt. Ich bin allerdings nicht sicher, ob man wirklich „echte“ Signale erkennen könnte, da die Geschwindigkeit des Programms auf 500Hz beschränkt ist (Intervall = 2ms).

Die Benutzeroberfläche

Selbstverständlich will ich Ihnen nicht vorschreiben, wie sie die Oberfläche gestalten sollen. Sie könnte allerdings in etwa so aussehen, wie auf dem Screenshot. 

Hinweis: Die blaue Linie auf dem Ausgabe-Bildschirm ist ein Line-Control, sie wird nicht im Programm gezeichnet!

Der Timer

Fangen wir einmal damit an. Wenn Sie wollen, können Sie gerne den VB-eigenen Timer verwenden. Ich mache es jedenfalls nicht. Bei mir habe ich den Timer aus dem Tipp !tip0101 verwendet. Er ist äusserst genau und recht einfach zu handhaben. Allerdings müssen Sie folgendes beachten:

· In der Timer-Prozedur (TimerClb) dürfen keine APIs ausser PostMessage aufgerufen werden!

· Intervalle unter einer Millisekunde sind gefährlich!

· Wenn der Timer beim Beenden des Programms nicht abgeschaltet wird, dann stürzt die IDE ab!

Scheinbar ist es auch nicht möglich, PostMessage einfach als API aufzurufen, da beim Aufruf jeweils LoadLibrary ausgeführt wird, was in der Exe zum Absturz führt. Stattdessen muss man die Typelibrary „PostMessageA“ unter Projekt -> Verweise einbinden und dann mit Call PostMessageA() aufrufen.

Soviel zur Theorie. Jetzt kommt die Praxis:

Zuerst sollten Sie ein neues Modul erstellen (Projekt -> Modul hinzufügen -> Modul). In dieses setzen Sie folgenden Code:

[code]

Option Explicit

Private Declare Function timeKillEvent Lib "winmm.dll" ( _

  ByVal uID As Long) As Long

Private Declare Function timeSetEvent Lib "winmm.dll" ( _

  ByVal uDelay As Long, _

  ByVal uResolution As Long, _

  ByVal lpFunction As Long, _

  ByVal dwUser As Long, _

  ByVal uFlags As Long) As Long

Private Const TIME_PERIODIC As Long = 1&

Public Busy As Boolean  'True, solange die FireTimer-Prozedur noch ausgeführt wird

Dim TimerID As Long

'Timer starten

Public Sub StartTimer(Interval As Long)

    TimerID = timeSetEvent(Interval, 10000, _

      AddressOf TimerClb, 0, TIME_PERIODIC)

End Sub

'Timer stoppen

Public Sub StopTimer()

    Call timeKillEvent(TimerID)

End Sub

'Timer-Ereignis

Public Sub TimerClb(ID As Long, Msg As Long, LngUser As Long, _

  Lng1 As Long, Lng2 As Long)

    'Timer auslösen (Aber nur, wenn die FireTimer-Prozedur fertig

    'abgearbeitet ist!)

    'Durch das Senden der Windows-Message wird bei der PicBox das

    'MouseMove-Ereignis ausgelöst.

    If Not Busy Then

        Call PostMessageA(frmMain.picTimer.hwnd, WM_MOUSEMOVE, 0&, 0&)

    End If

End Sub

[/code]

So, das hätten wir. Jetzt muss diese Message aber natürlich auch noch irgendwie abgefangen werden. Darum setzen wir auf die Form jetzt noch eine unsichtbare PicBox „picTimer“.

Allgemeine Deklarationen

Als nächstes brauchen wir nun noch mal ein Modul „basGlobal“, in das alle globalen Deklarationen kommen:

[code]

Option Explicit

'Vom Port lesen

Public Declare Function Inp Lib "inpout32.dll" Alias "Inp32" ( _

  ByVal PortAddress As Integer) As Integer

'In den Port schreiben

Public Declare Sub Out Lib "inpout32.dll" Alias "Out32" ( _

  ByVal PortAddress As Integer, ByVal Value As Integer)

'Pixel setzen

Public Declare Function SetPixel Lib "gdi32.dll" ( _

  ByVal hDC As Long, _

  ByVal X As Long, _

  ByVal Y As Long, _

  ByVal crColor As Long) As Long

'Bild verschieben/kopieren

Public Declare Function BitBlt Lib "gdi32.dll" ( _

  ByVal hDestDC As Long, _

  ByVal X As Long, _

  ByVal Y As Long, _

  ByVal nWidth As Long, _

  ByVal nHeight As Long, _

  ByVal hSrcDC As Long, _

  ByVal XSrc As Long, _

  ByVal YSrc As Long, _

  ByVal dwRop As Long) As Long

'Bild von Quelle kopieren

Public Const SRCCOPY As Long = &HCC0020

'Text ausgeben

Public Declare Function PrintA Lib "gdi32.dll" Alias "TextOutA" ( _

  ByVal hDC As Long, _

  ByVal X As Long, _

  ByVal Y As Long, _

  ByVal lpString As String, _

  ByVal nCount As Long) As Long

'Position des Cursors ermitteln

Public Declare Function GetCursorPos Lib "user32.dll" ( _

  ByRef lpPoint As POINTAPI) As Long

'Cursorposition in einem Control errechnen

Public Declare Function ScreenToClient Lib "user32.dll" ( _

  ByVal hwnd As Long, _

  ByRef lpPoint As POINTAPI) As Long

Public Type POINTAPI

    X As Long

    Y As Long

End Type

'Die Anzahl Millisekunden seit dem Booten von Windows herauskriegen

Public Declare Function GetTickCount Lib "kernel32.dll" () As Long

Public Const WM_MOUSEMOVE As Long = &H200

[/code]

Die Meisten sollten Sie eigentlich kennen. Die Einzige, die wohl eher noch unbekannt sein könnte, ist die ScreenToClient-Funktion, die die Bildschirmkoordinaten der Maus in Koordinaten auf einem Control umrechnen kann.

HWnd benötigt den hWnd des Controls und lpPoint die Bildschirmkoordinaten der Maus. Nach dem Aufruf beinhaltet lpPoint die Koordinaten der Maus im Control.

Der hauptsächliche Code in frmMain

Zum Schluss kommt jetzt noch die Form frmMain, die die ganze Oberfläche und den meisten Code beinhaltet.

Zuerst die allgemeinen Deklarationen:

[code]

Dim CurrentTime As Long     'Protokollierte Zeit in ms

Dim MouseTime As Long       'Zeit unter der Maus

Dim MouseOffset As Long     'Abstand der Maus zum rechten Rand (in ms)

Dim TPPX As Long            'TwipsPerPixelX

Dim TPPY As Long            'TwipsPerPixelY

Dim RunProtocolling As Boolean  'Läuft die Protokollierung?

Dim Interval As Long        'Intervall des Timers

Dim NoBitClick As Boolean   'True, wenn kein chkBit_Click-Ereignis auftreten soll

[/code]

Dann kommt die Startprozedur Form_Load und die Endprozedur Form_Unload:

[code]

Private Sub Form_Load()

    TPPX = Screen.TwipsPerPixelX

    TPPY = Screen.TwipsPerPixelY

    'Timer starten

    'Der Intervall darf nicht zu niedrig werden, da es sonst verschachtelte

    'Aufrufe von TimerClb gibt, was unweigerlich einen Absturz nach sich zieht!

    Interval = IIf(Val(txtInterval.Text) > 0, Val(txtInterval.Text), 20)

    Call StartTimer(Interval)

End Sub

'Vor dem Beenden MUSS der Timer gestoppt werden!

Private Sub Form_Unload(Cancel As Integer)

    Call StopTimer

End Sub

[/code]

Soweit so gut.

Jetzt ist der wichtigste Teil an der Reihe, nämlich die ganze Verwaltung des LPTs und die Ausgabe der Signale:

[code]

'Diese PicBox ist nur dazu da, ein Ereignis auszulösen, wenn der Timer

'gefeuert wird, da in der TimerClb NUR POSTMESSAGE und KEINE ANDEREN

'APIS aufgerufen werden dürfen!

Private Sub picTimer_MouseMove(Button As Integer, Shift As Integer, _

  X As Single, Y As Single)

    Call FireTimer

End Sub

'Timer wurde ausgelöst

Public Sub FireTimer()

    Dim i As Long

    Dim TW As Long

    Dim TH As Long

    Dim MousePos As POINTAPI    'Mausposition am Bildschirm

    Static FPS As Long          'Momentane FPS-Zahl

    Static oldTick As Long      'GetTickCount bei der letzten FPS-Aktualisierung

    Static PrevTick As Long     'GetTickCount beim letzten Timer-Aufruf

    Static OutputTick As Long   'GetTickCount bei der letzten Akt. der Ausgänge

    On Error GoTo ErrorHandler

    'Bitte nicht stören!

    Busy = True

    'Nur zeichnen, wenn RunProtocolling eingeschaltet ist!

    If RunProtocolling Then

        'Die PicBox ist jetzt 165 Pixel hoch, das ergibt 20 Pixel

        'pro Bit und 5 Pixel Rand.

        'VORSICHT: Einige Eingänge des LPTs sind invertiert! Mit XOr 120

        '          kann diese aufgehoben werden!

        'Autoredraw einschalten (Flackern unterdrücken)

        picDiagram.AutoRedraw = True

        'Erstmal die Beschriftungen löschen

        TW = picDiagram.TextWidth("Bit 0")

        TH = picDiagram.TextHeight("0")

        For i = 0 To 7

            picDiagram.Line (2, i * 20 + 2)-(2 + TW, i * 20 + 2 + TH), vbBlack, BF

        Next i

        'Dann das ganze Bild ein Pixel nach links schieben

        With picDiagram

            Call BitBlt(.hDC, 0, 0, .ScaleWidth, .ScaleHeight, .hDC, 1, 0, SRCCOPY)

        End With

        'Die Spalte ganz rechts löschen

        With picDiagram

            picDiagram.Line (.ScaleWidth - 1, 0)-(.ScaleWidth - 1, .ScaleHeight - 1), vbBlack

        End With

        'Jetzt wieder beschriften und die Linie weiterziehen

        For i = 0 To 7

            Call PrintA(picDiagram.hDC, 2, i * 20 + 2, "Bit " & i, 5)

            'Bei Null ist die Linie zwei Pixel dick, sonst vier

            With picDiagram

                Call SetPixel(.hDC, .ScaleWidth - 1, i * 20 + 18, vbGreen)

                Call SetPixel(.hDC, .ScaleWidth - 1, i * 20 + 19, vbGreen)

                If (Inp(CLng("&H" & UCase$(txtPortInput.Text))) Xor _

                  IIf(chkXOr128.Value, 128, 0)) And 2 ^ i Then

                    'Eins

                    Call SetPixel(.hDC, .ScaleWidth - 1, i * 20 + 16, vbGreen)

                    Call SetPixel(.hDC, .ScaleWidth - 1, i * 20 + 17, vbGreen)

                End If

            End With

        Next i

        'Zeit aktualisieren

        CurrentTime = CurrentTime + Interval

        'Autoredraw wieder ausschalten

        picDiagram.AutoRedraw = False

    End If

    lblTime.Caption = Format$(CurrentTime / 1000, "000.000s")

    'Zeit unter der Maus anzeigen

    '1 Pixel entspricht [Intervall]ms

    Call GetCursorPos(MousePos)

    Call ScreenToClient(picDiagram.hwnd, MousePos)

    'Befindet sich die Maus in der PicBox?

    If MousePos.X >= 0 And MousePos.X < picDiagram.ScaleWidth And _

      MousePos.Y >= 0 And MousePos.Y < picDiagram.ScaleHeight Then

        'Ja

        MouseOffset = (picDiagram.ScaleWidth - MousePos.X - 1) * Interval

        MouseTime = CurrentTime + MouseOffset

        lblMouseTime.Caption = Format$(MouseTime / 1000, "000.000s")

        lneMouseX.X1 = MousePos.X

        lneMouseX.X2 = MousePos.X

        lneMouseX.Y1 = 0

        lneMouseX.Y2 = picDiagram.ScaleHeight

        lneMouseX.Visible = True

    Else

        'Nein

        lblMouseTime.Caption = "000.000s"

        lneMouseX.Visible = False

    End If

    'FPS berechnen

    FPS = FPS + 1

    If Abs(oldTick - GetTickCount) >= 1000 Then

        lblFPS.Caption = Format$(FPS, "000")

        FPS = 0

        oldTick = GetTickCount

    End If

    'SPF berechnen

    lblSPF.Caption = Format$(Abs(PrevTick - GetTickCount) / 1000, "0.000")

    PrevTick = GetTickCount

    'Evtl. Ausgänge aktualisieren (alle 300ms)

    If Abs(OutputTick - GetTickCount) >= 300 Then

        OutputTick = GetTickCount

        Call Outputs

    End If

    Busy = False

    Exit Sub

ErrorHandler:

    Call MsgBox("Error: " & Err.Description)

    Call Err.Clear

End Sub

[/code]

Hier gibt es jetzt wohl einiges zu erklären:

Zuerst die MouseMove-Prozedur der unsichtbaren PicBox „picTimer“. Sie wird mit dem PostMessage-Aufruf ausgelöst, was den Aufruf anderer APIs möglich macht, da sie damit die Synchronisation verliert.

Dann zur FireTimer-Prozedur:

Zuerst werden einige Variablen deklariert, die wir später benötigen. Dann wird das Busy-Flag auf True gesetzt, was eine Verschachtelung der Aufrufe verhindert, die mit der Zeit einen Absturz nach sich ziehen würde.

Wenn dann RunProtocolling auf True gesetzt wurde (Geschieht in der cmdStart_Click-Prozedur), wird AutoRedraw der Ausgabe-PicBox eingeschaltet, was ein Flackern verhindert.

Dann werden die Beschriftungen „Bit 0“ bis „Bit 7“ mit schwarzen Rechtecken überschrieben, damit sie nicht auch nach links geschoben werden. Mittels BitBlt-API wird daraufhin das ganze Bild um ein Pixel nach links kopiert, damit ganz rechts wieder Platz für neue Werte frei wird, der gleich darauf gelöscht wird, damit nach einer Eins (dicke Linie) auch wirklich wieder eine Null angezeigt werden kann.

Nachher kommt die eigentliche Ausgabe des Diagramms:

Zuerst werden zwei Pixel gleich übereinander gezeichnet, die für eine Null benötigt werden. Wenn dann das aktuelle Bit eine Eins hat, dann werden darüber noch mal zwei Pixel gezeichnet, damit es eine Eins ist.

Wenn alles zeichnerische erledigt ist, dann kommt jetzt noch die Zeitmessung. Zuerst wird die aktuelle Zeit um den Intervall erhöht, damit sie auch bei verschiedenen Intervallen noch stimmt. Danach wird AutoRedraw wieder ausgeschaltet, damit es nicht flackert. Alles was jetzt noch kommt wird immer gemacht, also auch, wenn das Diagramm gar nicht gezeichnet wird.

Zuerst wird einmal die aktuelle Zeit angezeigt. Dann wird die Mausposition am Bildschirm gelesen und in Koordinaten auf der PicBox umgerechnet. Wenn sie sich wirklich innerhalb der PicBox befindet, dann wird das Offset (der Abstand der Maus vom rechten Rand) in Millisekunden berechnet und in MouseOffset gespeichert. Damit kann die absolute Zeit unter dem Cursor ausgerechnet werden.

Danach wird die Mauszeit ausgegeben und die Linie entsprechend verschoben.

Nach allen diesen Maus-Angelegenheiten kommt jetzt noch die Überprüfung der Timer-Genauigkeit dazu. Zuerst werden die FPS (Frames per second) berechnet, also wie viele Durchläufe pro Sekunde ausgeführt werden. Dann wird noch das Umgekehrte gemessen, also wie lange es von einem bis zum nächsten Aufruf der Prozedur dauert.

Schlussendlich wird noch in einem 300ms-Takt die Prozedur „Outputs()“ aufgerufen, die die Ausgabe auf den LPT ermöglicht.

Danach kommen noch Aufräumarbeiten, die eigentlich verständlich sein sollten.

Alles für die Eingänge...

...kommt dann. Also die Ereignisse beim Starten, bzw. Stoppen der Diagramm-Aufzeichnung und die Gültigkeitskontrolle einer Änderung des Intervalls.

[code]

'EINGÄNGE----------------------------------------------------------------------

'Timer starten

Private Sub cmdStart_Click()

    RunProtocolling = True

End Sub

'Timer stoppen

Private Sub cmdStop_Click()

    RunProtocolling = False

End Sub

'Zeit zurücksetzen

Private Sub cmdResetTime_Click()

    CurrentTime = -MouseOffset

End Sub

Private Sub txtInterval_Change()

    'Intervall wurde geändert

    'Wenn das Textfeld keinen numerischen Wert beinhaltet, wird als Intervall

    'einfach 20 angenommen

    Interval = Val(txtInterval.Text)

    If Interval = 0 Then Interval = 20

    'Timer aktualisieren

    Call StopTimer

    Call StartTimer(Interval)

    'Wenn der Intervall geändert wurde stimmt das Gezeichnete

    'natürlich NICHT MEHR! Deshalb die PicBox leeren.

    picDiagram.AutoRedraw = True

    Call picDiagram.Cls

    picDiagram.AutoRedraw = False

    Call picDiagram.Refresh

End Sub

[/code]

...und für die Ausgänge

Alles was jetzt noch fehlt ist die Steuerung der Ausgänge des LPTs. Zuerst kommt da die Prozedur „Outputs“, die alle 300ms aufgerufen wird und die Checkboxes entsprechend dem eingestellten Modus (Binärzähler, Blinken, Lauflicht, Zufall) verändert. Die Variable NoBitClick ist dazu da, ein Click-Ereignis bei den Bit-Checkboxes zu unterbinden, solange die Prozedur läuft, da sie sonst viel zu häufig auftreten würden.

Ansonsten sollte der Code recht verständlich sein. Es sind halt einfach einige Bit-Operationen, die man beherrschen sollte, wenn man mit dem LPT arbeiten will.

[code]

'AUSGÄNGE----------------------------------------------------------------------

'Ausgänge aktualisieren (Eingestellten Wert ausgeben)

Private Sub Outputs()

    Static LightCount As Long       'Zähler für's Lauflicht (0 - 7)

    Static BlinkState As Boolean    'Status des Blinklichtes (Ein/Aus)

    Dim i As Long

    Dim X As Long

    On Error Resume Next

    Call Randomize              'Zufallsgenerator initialisieren

    'Kein chkBit_Click-Ereignis auslösen!

    NoBitClick = True

    'Erstmal den Modus auswählen

    If chkOutputMode(0).Value Then

        'Binärzähler

        'Aktuellen Wert lesen

        For i = 0 To 7

            If chkBit(i).Value = 1 Then

                X = X + 2 ^ i

            End If

        Next i

        'Dann den Wert um Eins erhöhen, bzw. wieder bei Null anfangen

        'und ausgeben

        '(Das würde Klaus jetzt wohl alles gleich bitweise berechnen,

        'siehe AVB-Computer :D)

        X = X + 1

        If X > 255 Then X = 0

        For i = 0 To 7

            chkBit(i).Value = Abs((X And 2 ^ i) > 0)

        Next i

    ElseIf chkOutputMode(1).Value Then

        'Lauflicht

        'Wert erhöhen, bzw. wieder auf Null setzen

        LightCount = LightCount + 1

        If LightCount > 7 Then LightCount = 0

        'Dann den Wert ausgeben

        For i = 0 To 7

            chkBit(i).Value = Abs(LightCount = i)

        Next i

    ElseIf chkOutputMode(2).Value Then

        'Blinken

        'Wert umkeheren und ausgeben

        BlinkState = Not BlinkState

        For i = 0 To 7

            chkBit(i).Value = Abs(BlinkState)

        Next i

    ElseIf chkOutputMode(3).Value Then

        'Zufall

        'Zufällige Bits ein-/ausschalten

        For i = 0 To 7

            chkBit(i).Value = Abs(Rnd >= 0.5)

        Next i

    End If

    NoBitClick = False

    Call chkBit_Click(0)

    Call Err.Clear

End Sub

'Ein Bit wurde verändert -> Neu ausgeben!

Private Sub chkBit_Click(Index As Integer)

    Dim X As Long

    Dim i As Long

    On Error Resume Next

    If NoBitClick Then Exit Sub

    'Wert berechnen

    For i = 0 To 7

        'Kein Click-Ereignis auslösen

        chkBit(i).Enabled = False

        If chkBit(i).Value Then

            X = X + 2 ^ i

        End If

        chkBit(i).Enabled = True

    Next i

    'Und ausgeben

    Call Out(CLng("&H" & UCase$(txtPortOutput.Text)), X)

    Call Err.Clear

End Sub

'Es kann immer nur EIN Modus ausgewählt sein!

Private Sub chkOutputMode_Click(Index As Integer)

    Dim i As Long

    Static NoEvent As Boolean   'True, wenn das Click-Ereignis

                                'nicht ausgelöst werden soll

    If NoEvent Then Exit Sub

    'Die anderen Modis ausschalten

    For i = 0 To 3

        NoEvent = True

        If i <> Index Then

            chkOutputMode(i).Value = False

        End If

        NoEvent = False

    Next i

End Sub

[/code]

Schlusswort

Das war’s jetzt eigentlich von meiner Seite. Ich hoffe, damit sind nun einige Fragen geklärt, die schon einige Male im Hardware-Forum gestellt wurden.

Nun kann ich Ihnen nur noch raten, einfach ein bisschen auszuprobieren. Und keine Angst: So empfindlich, wie man vielleicht denkt, ist der LPT eigentlich gar nicht. Trotzdem sollten Sie es natürlich nicht übertreiben und gleich alle Ausgänge kurzschliessen oder dergleichen.

Ich bin gerade dabei, das Programm „Port-Diagram“ das wir eben erstellt haben, noch etwas zu verbessern, damit man auch Datenfolgen speichern und wiedergeben kann. Ich wollte damit eigentlich eine kleine IR-Fernsteuerung realisieren, aber anscheinend ist ein Intervall von 2ms dafür wirklich zu langsam. Wahrscheinlich müsste man dazu eine Schleife verwenden, deren Geschwindigkeit stabilisiert wird. Das soll jetzt allerdings nicht mehr Thema dieses Tutorials sein, da es nicht mehr so viel mit dem LPT zu tun hat.

Noch ein Hinweis:

Scheinbar funktioniert die Dll in der IDE nicht, wenn sie sich im System-Verzeichnis und im Programmverzeichnis befindet!

Falls Sie Fragen oder Anregungen bezüglich des Tutorials oder des Programms haben, zögern Sie nicht, mir eine E-Mail (philipp_burch@bluemail.ch) zu senden oder mich per MSN (siehe E-Mail-Adresse) oder ICQ (199-644-984) zu kontaktieren!

Mit freundlichen Grüssen

  Philipp „Phip“ Burch

Dateien:

Archiv MoniStandby.zip

Archiv Port-Diagram.zip

InpOut32.dll

Tutorial - Stefan Maag.txt

